
ЦЕЛЬ  РАБОТЫ: Закрепление знаний об  объектно-ориентированном программировании (ООП), разработка объектов и объектно-ориентированных программ.  
ОСНОВНЫЕ   СВЕДЕНИЯ

Объектно-ориентированное программирование базируется на трех основных принципах: инкапсуляция, наследование, полиморфизм.

Инкапсуляция – это объединение в одном объекте данных и методов работы с этими данными. 

Наследование – использование в объектах-потомках как всех свойств и методов объектов-предков, так и новых методов.

Полиморфизм – зависимость одноименных методов работы с данными от иерархии объектов, которым принадлежат эти методы.

Ниже приведен пример программы с графическими объектами: позицией, точкой, окружностью.

uses dos,crt,graph;

type

Location = object  

{объект - позиция, предок}
x,y :integer;

procedure Init(InitX,InitY : integer);

function GetX : integer;

function GetY : integer;

end;

Point = object(Location) 
{объект - точка, потомок позиции}
visible : boolean;

constructor Init(InitX, InitY : integer);

procedure Show ;virtual; {виртуальный метод}

procedure Hide; virtual; {виртуальный метод}

function IsVisible : boolean;

procedure MoveTo(NewX, NewY : integer);

end;

Circle = object(Point)

{объект-окружность, потомок точки}
radius : integer;

constructor Init(InitX, InitY : integer;

                   InitRadius : integer);

procedure Show; virtual; {как потомок виртуального}
procedure Hide; virtual; {как потомок виртуального}
procedure Expand(ExpandBy : integer);

procedure Contract(ContractBy : integer);

procedure MoveTo(NewX, NewY : integer);

end;

{После описания объектов следуют описания методов объектов}
   procedure Location.Init(InitX,InitY : integer);

   begin

     x:=InitX;

     y:=InitY;

   end;
   function Location.GetX : integer;

   begin

     GetX:=x;

   end;

   function Location.GetY : integer;

   begin

     GetY:=y;

   end;

  constructor Point.Init(InitX, InitY : integer);

  begin

    Location.Init(InitX, InitY);

    visible:=false;

  end;
  procedure Point.Show ;

  begin

    visible:=true;

    putpixel(x,y,getcolor);

  end;
  procedure Point.Hide ;

  begin

    visible:=false;

    putpixel(x,y,getbkcolor);

  end;
  function Point.IsVisible;

  begin

    IsVisible:=visible;

  end;
  procedure Point.MoveTo(NewX,NewY:integer);

  begin

    Hide;

    x:=NewX;

    y:=NewY;

    Show;

  end;
  constructor Circle.Init(InitX, InitY, InitRadius : integer);

  begin

    Point.Init(InitX,InitY);

    radius:=InitRadius;

  end;

  procedure Circle.Show;

  begin

    visible:=true;

    Graph.Circle(x,y,radius);

  end;
  procedure Circle.Hide;

  var

  tempcolor:word;

  begin

    tempcolor:=getcolor;


{запомнить текущий цвет}
    setcolor(getbkcolor); 

{стирание-цвет фона!   }
    visible:=false;

    Graph.Circle(x,y,radius);

    setcolor(tempcolor); 

{восстановление цвета  }
  end;
  procedure Circle.Expand(ExpandBy : integer);

  begin

    Hide;

    radius:=radius+ExpandBy;

    if radius<0 then radius:=0;

    Show;

  end;
  procedure Circle.Contract(ContractBy : integer);

  begin

    Expand(-ContractBy); 

{использовать готовый метод!}
  end; 

  procedure Circle.MoveTo(NewX, NewY : integer);

  begin

    Hide;

    x:=NewX;

    y:=NewY;

    Show;

  end;

 var

 driver,mode:integer;

 MyCircle:Circle;

 begin 





{основная программа}
    {инициализация графического режима}

    driver:=9; mode:=2;

    initgraph(driver,mode,'..\bgi');

    MyCircle.Init(100,200,50); 
{создать окружность}
    MyCircle.Show; 



{показать окружность}
    readkey; 




{ждать нажатия клавиши}
    MyCircle.MoveTo(200,250); 
{передвинуть окружность}
    readkey;

    MyCircle.Expand(50); 

{увеличить радиус на 50}
    readkey;

    MyCircle.Contract(75); 

{уменьшить радиус на 75}

    readkey;

    closegraph;

  end.
ВЫПОЛНЕНИЕ РАБОТЫ

Внимательно ознакомьтесь с текстом приведенной программы на применение объектов, наберите и исполните ее. 

Постарайтесь выяснить разницу между методами MoveTo (Переместить) для объектов Point и Circle. Формально запись процедур одинакова: стирание, присваивание новых координат, рисование. Однако рисование окружности (Circle.Show) отличается от рисования точки (Point.Show). 
Ознакомившись с примером программы с объектами (2 варианта), разработать самостоятельно программу с определением иерархии объектов и использованием наследования и полиморфизма. При этом следует иметь в виду, что объекты могут быть не только графические.





































